HEWLETT-PACKARD

HP-34C

STUDENT ENGINEERING APPLICATIONS

HEWLETT PACKARD

HP-34C

Student Engineering Applications

August, 1979

00034-90035

Printed in U.S.A.

Introduction

This HP-34C Applications book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered in engineering.

They will provide you with immediate capabilities in your everyday calculations and you may find them useful as guides to programming techniques for writing your own customized software.

You will find general information on how to key in and run programs under ''A Word about Program Usage'' in the Applications book you received with your calculator.

We hope that this Student Engineering book will be a valuable tool in your work and would appreciate your comments about it.

Contents

Electrical Engineering

Ohm's Law

This program provides interchangeable solutions for the Ohm's Law relations. Specifically, one may solve for:

- Resistance and power dissipation given voltage and current.
- Current and power dissipation given voltage and resistance.
- Voltage and power dissipation given current and resistance.
- Current and resistance given voltage and power dissipation.
- Voltage and resistance given current and power dissipation.
- Voltage and current given resistance and power dissipation.

Note:

This program can also be loaded and run with either the Reactance Chart program or the Series-Parallel Resistor Adding routine.

Formulas used:

$$V = IR$$

$$P = I^2R$$

Where

V = voltage

I = current

R = resistance

P = power dissipation

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL A	001–25, 13, 11
RCL 0	002– 24 0
g x≠0	003– 15 61
GTO 1	004– 22 1
RCL 1	005– 24 1
g x≠0	006– 15 61
GTO 2	007– 22 2
RCL 3	008– 24 3
RCL 2	009– 24 2
+	010– 71
f √x	011– 14 3
STO 1	012– 23 1
GTO 3	013– 22 3
h LBL 1	014–25, 13, 1
RCL 1	015– 24 1
g x≠0	016– 15 61
GTO 4	017– 22 4
RCL 2	018– 24 2
g x≠0	019– 15 61
GTO 0	020– 22 0
RCL 3	021– 24 3
RCL 0	022– 24 0
+	023– 71
STO 1	024– 23 1
GTO 4	025– 22 4
h LBL 2	026–25, 13, 2
RCL 2	027– 24 2
g x≠0	028– 15 61

KEY ENTRY	DISPLAY
GTO 3	029– 22 3
RCL 3	030– 24 3
RCL 1	031– 24 1
g x^2	032– 15 3
+	033– 71
STO 2	034– 23 2
h LBL 3	035–25, 13, 3
RCL 1	036– 24 1
RCL 2	037– 24 2
×	038– 61
STO 0	039– 22 0
GTO 5	040– 22 5
h LBL 4	041–25, 13, 4
RCL 0	042– 24 0
RCL 1	043– 24 1
+	044– 71
STO 2	045– 23 2
GTO 5	046– 22 5
h LBL 0	047–25, 13, 0
RCL 0	048– 24 0
RCL 2	049– 24 2
+	050– 71
STO 1	051– 23 1
h LBL 5	052–25, 13, 5
RCL 1	053– 24 1
RCL 0	054– 24 0
×	055– 61
STO 3	056– 23 3
h RTN	057– 25 12

REGISTERS			I Unused
R_0 *V*	R_1 *I*	R_2 *R*	R_3 *P*
R_4-$R_{.9}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store any two of the			
	following:			
	Voltage, *V*, if known	*V*, volts	STO 0	
	Current, *I*, if known	*I*, amps	STO 1	
	Resistance, *R*, if known	*R*, ohms	STO 2	
	Power dissipation, *P*,			
	if known	*P*, watts	STO 3	
3	Clear the two unknowns:			
	If *V* is unknown	0	STO 0	
	If *I* is unknown	0	STO 1	
	If *R* is unknown	0	STO 2	
	If *P* is unknown	0	STO 3	
4	Calculate unknowns.		A	*P*, watts
5	Recall desired values.		RCL 0	*V*, volts
			RCL 1	*I*, amps
			RCL 2	*R*, ohms
			RCL 3	*P*, watts

Example 1:

V = 43.2 volts, I = .1 amps; calculate R and P:

Keystrokes:	Display:	
43.2 [STO] 0		
.1 [STO] 1		
0 [STO] 2		
[STO] 3 [A]	4.3200	(P, watts)
[RCL] 2	432.0000	(R, ohms)

Example 2:

After Example 1 let V = 43.2 volts, R = 430; calulate I and P:

Keystrokes:	Display:	
430 [STO] 2		
0 [STO] 1		
[STO] 3 [A]	4.3401	(P, watts)
[RCL] 1	0.1005	(I, amps)

(Notice that since V was unchanged it did not need to be stored.)

Reactance Chart

This program calculates inductive and capacitive reactance and provides interchangeable solutions between frequency, inductance and capacitance at resonance.

- Since, at resonance, capacitive and inductive reactances are equal, they will both be termed, simply, reactance.
- This program can be loaded and run with the Ohm's Law program.

Formulas Used:

$$f = \frac{1}{2\pi\sqrt{LC}}$$

$$X = \frac{1}{2\pi fC}$$

where:

f = resonant frequency (hertz)
L = inductance (henrys)
C = capacitance (farads)
X = reactance (ohms)

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL B	001–25, 13, 12
RCL 4	002– 24 4
g x≠0	003– 15 61
GTO 0	004– 22 0
RCL 5	005– 24 5
RCL 6	006– 24 6
×	007– 61
f √x	008– 14 3
GSB 3	009– 13 3
STO 4	010– 23 4
GTO 2	011– 22 2
h LBL 0	012–25, 13, 0
GSB 3	013– 13 3
g x^2	014– 15 3
RCL 5	015– 24 5
g x≠0	016– 15 61
GTO 1	017– 22 1
g R↓	018– 15 22
RCL 6	019– 24 6
÷	020– 71
STO 5	021– 23 5
GTO 2	022– 22 2
h LBL 1	023–25, 13, 1
÷	024– 71
STO 6	025– 23 6
h LBL 2	026–25, 13, 2
RCL 4	027– 24 4
RCL 6	028– 24 6
×	029– 61
h LBL 3	030–25, 13, 3
h π	031– 25 73
×	032– 61
2	033– 2
×	034– 61
h 1/x	035– 25 2
STO 7	036– 23 7
h RTN	037– 25 12

REGISTERS			I Unused
R_0	R_1	R_2	R_3
R_4 *f*	R_5 *L*	R_6 *C*	R_7 *X*
R_8–$R_{.9}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store any two of the			
	following:			
	Frequency, *f*, if known	*f*, hertz	[STO] 4	
	Inductance, *L*, if known	*L*, henrys	[STO] 5	
	Capacitance, *C*, if known	*C*, farads	[STO] 6	
3	Clear the unknown:			
	If *f* is unknown	0	[STO] 4	
	or, if *L* is unknown	0	[STO] 5	
	or, if *C* is unknown	0	[STO] 6	
4	Calculate the unknown and *X*,			
	the reactance.		[B]	*X*, ohms
5	Recall desired values.		[RCL] 4	*f*, hertz
			[RCL] 5	*L*, henrys
			[RCL] 6	*C*, farads
			[RCL] 7	*X*, ohms

Example 1:

$L = .1$ mh, $C = .2\ \mu$f; calculate f and X:

Keystrokes:	Display:	
.1 [EEX] [CHS] 3 [STO] 5		
.2 [EEX] [CHS] 6 [STO] 6		
0 [STO] 4 [B]	22.3607	(X, ohms)
[RCL] 4	35,588.1272	(f, hertz)

Example 2:

$f = 100$, $C = .1\ \mu$f; calculate X:

Keystrokes:	Display:	
100 [STO] 4		
.1 [EEX] [CHS] 6 [STO] 6		
0 [STO] 5 [B]	15,915.4943	(X, ohms)

Impedance of a Ladder Network

This program computes the input impedance of an arbitrary ladder network. Elements are added one at a time starting from the right. The first element must be in parallel.

Suppose we have a network whose input admittance is Y_{in}. Adding a shunt R, L or C, the input admittance becomes

$$Y_{\text{new}} = \begin{cases} Y_{\text{in}} + \left(\dfrac{1}{R_{\text{p}}} + \text{j}0 \right) \\[2ex] Y_{\text{in}} + \left(0 - \text{j}\dfrac{1}{\omega L_{\text{p}}} \right) \\[2ex] Y_{\text{in}} + (0 + \text{j}\,\omega C_{\text{p}}) \end{cases}$$

Adding a series R, L or C, we have

$$Y_{\text{new}} = \begin{cases} \left(\dfrac{1}{Y_{\text{in}}} + (R_{\text{s}} + \text{j}0) \right)^{-1} \\[2ex] \left(\dfrac{1}{Y_{\text{in}}} + (0 + \text{j}\,\omega L_{\text{s}}) \right)^{-1} \\[2ex] \left(\dfrac{1}{Y_{\text{in}}} + \left(0 - \text{j}\dfrac{1}{\omega C_{\text{s}}} \right) \right)^{-1} \end{cases}$$

The program converts this admittance to an impedance for display.

Note:

An erroneous entry may be corrected by entering the negative of the incorrect value.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL A	001–25, 13, 11
f CLEAR REG	002– 14 33
2	003– 2
×	004– 61
h π	005– 25 73
×	006– 61
STO 0	007– 23 0
h CF 0	008–25, 61, 0
h RTN	009– 25 12
h LBL B	010–25, 13, 12
h SF 0	011–25, 51, 0
h RTN	012– 25 12
h LBL 1	013–25, 13, 1
h 1/x	014– 25 2
h F? 0	015–25, 71, 0
h 1/x	016– 25 2
0	017– 0
GTO 9	018– 22 9
h LBL 2	019–25, 13, 2
h F? 0	020–25, 71, 0
GTO 6	021– 22 6
h LBL 5	022–25, 13, 5
RCL 0	023– 24 0
×	024– 61
h 1/x	025– 25 2
CHS	026– 32
0	027– 0
x⇄y	028– 21

KEY ENTRY	DISPLAY
GTO 9	029– 22 9
h LBL 3	030–25, 13, 3
h F? 0	031–25, 71, 0
GTO 5	032– 22 5
h LBL 6	033–25, 13, 6
RCL 0	034– 24 0
×	035– 61
0	036– 0
x⇄y	037– 21
h LBL 9	038–25, 13, 9
RCL 2	039– 24 2
RCL 1	040– 24 1
h F? 0	041–25, 71, 0
GSB 0	042– 13 0
f R↑	043– 14 22
+	044– 51
g R↓	045– 15 22
+	046– 51
f R↑	047– 14 22
h F? 0	048–25, 71, 0
GSB 0	049– 13 0
STO 1	050– 23 1
x⇄y	051– 21
STO 2	052– 23 2
x⇄y	053– 21
GSB 0	054– 13 0
h CF 0	055–25, 61, 0
g →P	056– 15 4
h RTN	057– 25 12

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
h LBL 0	058– 25, 13, 0	CHS	062– 32
g →P	059– 15 4	x⇄y	063– 21
h 1/x	060– 25 2	f →R	064– 14 4
x⇄y	061– 21	h RTN	065– 25 12

REGISTERS			I Unused
R_0 ω	R_1 R_e $[Y_{in}]$	R_2 I_m $[Y_{in}]$	R_3–$R_{.9}$ Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Initialize.	f, hertz	A	ω, rad/sec
3	Input elements (first must			
	be parallel).			
3a	A parallel element.	R, ohms	GSB 1	$\lvert Z_{in}\rvert$, ohms
	or,	L, henrys	GSB 2	$\lvert Z_{in}\rvert$, ohms
	or,	C, farads	GSB 3	$\lvert Z_{in}\rvert$, ohms
3b	A series element.	R, ohms	B GSB 1	$\lvert Z_{in}\rvert$, ohms
	or,	L, henrys	B GSB 2	$\lvert Z_{in}\rvert$, ohms
	or,	C, farads	B GSB 3	$\lvert Z_{in}\rvert$, ohms
4	Repeat steps 3a or 3b for			
	next element.			
	(Optional): Whenever $\lvert Z_{in}\rvert$			
	is displayed the angle can be			
	found in the Y-register.		x⇄y	$\measuredangle Z_{in}$, deg

Example:

$f = 4$ MHz



Keystrokes:	**Display:**	
4 [EEX] 6 [A]		
50 [GSB] 1	50.0000	($\lvert Z_{in} \rvert$, ohms)
2400 [EEX] [CHS]		
12 [GSB] 3	15.7362	($\lvert Z_{in} \rvert$, ohms)
[x⇄y]	–71.6559	($\measuredangle Z_{in}$, deg)
2.56 [EEX] [CHS]		
6 [B] [GSB] 2	49.6509	($\lvert Z_{in} \rvert$, ohms)
[x⇄y]	84.2754	($\measuredangle Z_{in}$, deg)
796 [EEX] [CHS]		
12 [GSB] 3	497.6942	($\lvert Z_{in} \rvert$, ohms)
[x⇄y]	0.9840	($\measuredangle Z_{in}$, deg)

Series-Parallel Resistor Adding and Standard Resistance Values

This program will add series and parallel resistors, providing full use of the stack for intermediate answers. It also calculates the closest standard 5%, 10% or 20% resistor values.

The resistor adding and standard value routines are completely independent and either can be loaded and used without the other. The resistor adding routine is line 001-007 and the standard values routine is line 008-093.

The resistor adding routine (line 001-007) can also be loaded and run with the Ohm's Law program.

Note that capacitors in parallel and series can be substituted for resistors in series and parallel, respectively.

Reference for Standard Resistance Values:

International Telephone and Telegraph Corp., *Reference Data for Radio Engineers,* fourth edition, p. 78.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL B	001–25, 13, 12
h 1/x	002– 25 2
x⇄y	003– 21
h 1/x	004– 25 2
+	005– 51
h 1/x	006– 25 2
h RTN	007– 25 12
h LBL 1	008–25, 13, 1
1	009– 1
2	010– 2
0	011– 0
÷	012– 71
g 10ˣ	013– 15 2
STO 2	014– 23 2
g R↓	015– 15 22
h RTN	016– 25 12
h LBL A	017–25, 13, 11
f LOG	018– 14 2
ENTER↑	019– 31
h INT	020– 25 32
STO 4	021– 23 4
−	022– 41
1	023– 1
+	024– 51
g 10ˣ	025– 15 2
STO 3	026– 23 3
1	027– 1
STO − 4	028–23, 41, 4
1	029– 1
0	030– 0
STO 5	031– 23 5
h LBL 0	032–25, 13, 0
RCL 3	033– 24 3
RCL 5	034– 24 5
f x>y	035– 14 51
GTO 9	036– 22 9
RCL 2	037– 24 2
×	038– 61
STO 5	039– 23 5
GTO 0	040– 22 0
h LBL 9	041–25, 13, 9
GSB 8	042– 13 8
STO 7	043– 23 7
RCL 5	044– 24 5
RCL 2	045– 24 2
÷	046– 71
GSB 8	047– 13 8
STO 6	048– 23 6
RCL 7	049– 24 7
×	050– 61
f √x	051– 14 3
RCL 3	052– 24 3
f x≤y	053– 14 41
GSB 7	054– 13 7
RCL 7	055– 24 7
RCL 4	056– 24 4
g 10ˣ	057– 15 2

KEY ENTRY	DISPLAY
×	058– 61
h RTN	059– 25 12
h LBL 8	060–25, 13, 8
•	061– 73
5	062– 5
+	063– 51
h INT	064– 25 32
STO 6	065– 23 6
2	066– 2
6	067– 6
f x>y	068– 14 51
GTO 6	069– 22 6
4	070– 4
7	071– 7
RCL 6	072– 24 6
f x>y	073– 14 51
GTO 3	074– 22 3
1	075– 1

KEY ENTRY	DISPLAY
+	076– 51
h RTN	077– 25 12
h LBL 3	078–25, 13, 3
8	079– 8
3	080– 3
RCL 6	081– 24 6
f x≠y	082– 14 61
h RTN	083– 25 12
8	084– 8
2	085– 2
h RTN	086– 25 12
h LBL 7	087–25, 13, 7
RCL 6	088– 24 6
STO 7	089– 23 7
h RTN	090– 25 12
h LBL 6	091–25, 13, 6
RCL 6	092– 24 6
h RTN	093– 25 12

REGISTERS			I Unused
R_0	R_1	R_2 Step size	R_3 Normal *R*
R_4 Exp of *R*	R_5 This step	R_6 Temp	R_7 Temp
R_8–$R_{.5}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	To find standard resistance			
	values go to step 7.			
3	Key in resistor value.	R, ohms	[ENTER↑]	
4	Key in next resistor value.	R_i, ohms		
5	To add:			
5a	Parallel, or		[B]	R, ohms
5b	Series		[+]	R, ohms
6	To add other resistors go to			
	step 3, 4, or 5. For standard			
	resistance value go to step 7.			
7	Enter tolerance (5, 10 or 20			
	only). This only needs to be			
	done the first time, or whenever			
	a change is desired.	T%	[GSB] 1	
8	Enter non-standard resistor			
	size and calculate nearest			
	standard value.	R, ohms	[A]	R, ohms
9	For more calculations go to			
	step 3, 4, 7 or 8.			

Example 1:



Determine resistance from A to B.

Keystrokes:	**Display:**	
680 [ENTER↑]		
120 [B]		
330 [+]		
680 [+]		
220 [B]		
680 [ENTER↑]		
470 [B]		
[+]	461.5767	(R_{AB}, ohms)

Example 2:

Find the nearest 10% standard value resistor that will replace the circuit in Example 1, assuming the results from Example 1 still remain in the display.

Keystrokes:	**Display:**	
10 [GSB] 1 [A]	470.0000	(R, ohms)

Thermal and Transport Science

Ideal Gas Equation of State

Many gases obey the ideal gas laws quite closely at reasonable temperatures and pressures. This program calculates any one of the four variables when data for the other three and the universal gas constant are entered.

Equation:

$$PV = n\,RT$$

where:

P is the absolute pressure

V is the volume

n is the number of moles present

R is the Universal Gas Constant

T is the absolute temperature

Table 1

Values of the Universal Gas Constant

Value of R	Units of R	Units of P	Units of V	Units of T
8.314	N-m/g mole-K	N/m^2	m^3/g mole	K
83.14	cm^3-bar/g mole-K	bar	cm^3/g mole	K
82.05	cm^3-atm/g mole-K	atm	cm^3/g mole	K
0.08205	ℓ-atm/g mole-K	atm	ℓ/g mole	K
0.7302	atm-ft^3/lb mole-°R	atm	ft^3/lb mole	°R
10.73	psi-ft^3/lb mole-°R	psi	ft^3/lb mole	°R
1545	psf-ft^3/lb mole-°R	psf	ft^3/lb mole	°R

Remarks:

- At low temperatures or high pressures the ideal gas law does not represent the behavior of real gases.
- The value of R used must be compatible with the units of P, V, T.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	000–	GSB 9	020– 13 9
h LBL 1	001–25, 13, 1	RCL 3	021– 24 3
GSB 8	002– 13 8	÷	022– 71
RCL 2	003– 24 2	STO 4	023– 23 4
÷	004– 71	h RTN	024– 25 12
STO 1	005– 23 1	h LBL 8	025–25, 13, 8
h RTN	006– 25 12	RCL 3	026– 24 3
h LBL 2	007–25, 13, 2	RCL 0	027– 24 0
GSB 8	008– 13 8	×	028– 61
RCL 1	009– 24 1	RCL 4	029– 24 4
÷	010– 71	×	030– 61
STO 2	011– 23 2	h RTN	031– 25 12
h RTN	012– 25 12	h LBL 9	032–25, 13, 9
h LBL 3	013–25, 13, 3	RCL 1	033– 24 1
GSB 9	014– 13 9	RCL 2	034– 24 2
RCL 5	015– 24 5	×	035– 61
÷	016– 71	RCL 0	036– 24 0
STO 3	017– 23 3	÷	037– 71
h RTN	018– 25 12	h RTN	038– 25 12
h LBL 5	019–25, 13, 5		

REGISTERS			I Unused
R_0 R	R_1 P	R_2 V	R_3 n
R_4 T	R_5–$R_{.9}$ Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store *R* in appropriate units.	*R*	STO 0	*R*
3	Store any three knowns:			
	Pressure, *P*, if known	*P*	STO 1	
	Volume, *V*, if known	*V*	STO 2	
	Number of moles, *n*,			
	if known	*n*	STO 3	
	Temperature, *T*, if known	*T* (K or °R)	STO 4	
4	Solve for the unknown:			
	If *P* is unknown		GSB 1	*P*
	or, if *V* is unknown		GSB 2	*V*
	or, if *n* is unknown		GSB 3	*n*
	or, if *T* is unknown		GSB 4	*T* (K or °R)

Example 1:

.63 moles of air are enclosed in 25,000 cm^3 of space at 1.200 K. What is the pressure in bars? In atmospheres?

Keystrokes:	Display:	
83.14 STO 0		
25000 STO 2		
.63 STO 3		
1200 STO 4		
GSB 1	2.5142	(*P*, bars)
82.05 STO 0		
GSB 1	2.4812	(*P*, atm)

Example 2:

What is the specific volume (ft^3/lb) of a gas at atmospheric pressure and a temperature of 513°R? The molecular weight is 29 lb/lb−mole.

Keystrokes:	Display:	
.7302 [STO] 0		
1 [STO] 1		
29 [h] [1/x] [STO] 3		
513 [STO] 4		
[GSB] 2	12.9170	(V, ft^3/lb)
What is the density?		
[h] [1/x]	0.0774	(ρ, lb/ft^3)
What is the density at 1.32 atm and 555 °R?		
1.32 [STO] 1		
555 [STO] 4		
[GSB] 2 [h] [1/x]	0.0945	(ρ, lb/ft^3)

Conduit Flow

This program solves for the average velocity, or pressure drop for viscous, incompressible flow in conduits. Flow must be turbulent with Reynolds number greater than 4000.

Equations:

$$v^2 = \frac{\Delta P/\rho}{2\left(f\,\frac{L}{D} + \frac{K_T}{4}\right)}$$

$$0 = 1.737 \ln \frac{D}{\epsilon} + 2.28 - 1.737 \ln\left(4.67\,\frac{\frac{D}{\epsilon}}{\mathrm{Re}\,\sqrt{f}} + 1\right) - \frac{1}{\sqrt{f}}$$

is solved for $\frac{1}{\sqrt{f}}$ by the use of [SOLVE] with

$$1.737 \ln \frac{D}{\epsilon} + 2.28 \text{ and } \frac{1}{2}\left(1.737 \ln \frac{D}{\epsilon} + 2.28\right)$$

used as initial guesses.

where:

Re is the Reynolds number, defined as $\frac{D\mathrm{v}}{\nu}$

D is the pipe diameter;

ϵ is the dimension of irregularities in the conduit surface (see table 2);

f is the Fanning friction factor for conduit flow;

ΔP is the pressure drop along the conduit;

ρ is the density of the fluid;

ν is the kinematic viscosity of the fluid;

L is the conduit length;

v is the average fluid velocity;

K_T is the total of the applicable fitting coefficients in table 1.

Table 1
Fitting Coefficients

Fitting	*K*
Globe valve, wide open	7.5—10
Angle valve, wide open	3.8
Gate valve, wide open	0.15—0.19
Gate valve, ¾ open	0.85
Gate valve, ½ open	4.4
Gate valve, ¼ open	20
90° elbow	0.4—0.9
Standard 45° elbow	0.35—0.42
Tee, through side outlet	1.5
Tee, straight through	.4
180° bend	1.6
Entrance to circular pipe	0.25—0.50
Sudden expansion	$(1 - A_{up}/A_{dn})^{2*}$
Acceleration from $v = 0$ to $v = v_{entrance}$	1.0

* A_{up} is the upstream area and A_{dn} is the downstream area.

Table 2
Surface Irregularities

Material	ϵ (feet)	ϵ (meters)
Drawn or Smooth Tubing	5.0×10^{-6}	1.5×10^{-6}
Commericial Steel or Wrought Iron	1.5×10^{-4}	4.6×10^{-5}
Asphalted Cast Iron	4.0×10^{-4}	1.2×10^{-4}
Galvanized Iron	5.0×10^{-4}	1.5×10^{-4}
Cast Iron	8.3×10^{-4}	2.5×10^{-4}
Wood Stave	6.0×10^{-4} to 3.0×10^{-3}	1.8×10^{-4} to 9.1×10^{-4}
Concrete	1.0×10^{-3} to 1.0×10^{-2}	3.0×10^{-4} to 3.0×10^{-3}
Riveted Steel	3.0×10^{-3} to 3.0×10^{-2}	9.1×10^{-4} to 9.1×10^{-3}

Reference:

Welty, Wicks, Wilson; *Fundamentals of Momentum, Heat and Mass Transfer*, John Wiley and Sons, Inc., 1969.

Remarks:

- The solution requires an iterative procedure. The time solution will range from 10 seconds for ΔP, to a minute or more for V. The display setting is used to determine the accuracy of f via [CLX]. For more information on this see the Owner's Handbook. The accuracy of V is determined by checking the % change of V between successive iterations. When the % change is less than 10^{-3}, iteration is completed. Accuracy can be increased at the expense of running time by inserting a smaller number at lines 021-023.
- If the conduit is not circular, an equivalent diameter may be calculated using the formula below:

$$D_{\mathrm{eq}} = 4\,\frac{\text{cross sectional area}}{\text{wetted perimeter}}$$

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL B	001–25, 13, 12
h CF 0	002–25, 61, 0
GSB 3	003– 13 3
RCL 0	004– 24 0
g x^2	005– 15 3
×	006– 61
RCL 5	007– 24 5
×	008– 61
STO 1	009– 23 1
GTO 2	010– 22 2
h LBL A	011–25, 13, 11
h SF 0	012–25, 51, 0
GSB 3	013– 13 3
STO f I	014–23, 14, 23
h LBL 1	015–25, 13, 1
GSB 4	016– 13 4
ENTER↑	017– 31
f x⇄I	018– 14 21
h Δ%	019– 25 7
h ABS	020– 25 34
EEX	021– 33
3	022– 3
CHS	023– 32
f x≤y	024– 14 41
GTO 1	025– 22 1
RCL 0	026– 24 0
h LBL 2	027–25, 13, 2
RCL 9	028– 24 9

KEY ENTRY	DISPLAY
h 1/x	029– 25 2
g x^2	030– 15 3
STO 9	031– 23 9
g R↓	032– 15 22
h RTN	033– 25 12
h LBL 3	034–25 13 3
RCL 6	035– 24 6
RCL 7	036– 24 7
÷	037– 71
STO • 1	038– 23 .1
f LN	039– 14 1
1	040– 1
•	041– 73
7	042– 7
3	043– 3
7	044– 7
STO • 2	045– 23 .2
×	046– 61
2	047– 2
•	048– 73
2	049– 2
8	050– 8
+	051– 51
STO • 0	052– 23 .0
STO 9	053– 23 9
h F? 0	054–25, 71, 0
GTO 5	055– 22 5
h LBL 4	056–25, 13, 4
RCL 0	057– 24 0

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
RCL 6	058– 24 6	h F? 0	087– 25, 71, 0
×	059– 61	GTO 6	088– 22 6
RCL 4	060– 24 4	h RTN	089– 25 12
÷	061– 71	h LBL 6	090– 25, 13, 6
STO 8	062– 23 8	÷	091– 71
RCL 9	063– 24 9	f √x	092– 14 3
2	064– 2	STO 0	093– 23 0
÷	065– 71	h RTN	094– 25 12
RCL 9	066– 24 9	h LBL 0	095– 25 13 0
f SOLVE 0	067– 14, 73, 0	1	096– 1
GTO 5	068– 22 5	f x≤y	097– 14 41
0	069– 0	g R↓	098– 15 22
÷	070– 71	ENTER↑	099– 31
h LBL 5	071– 25, 13, 5	ENTER↑	100– 31
STO 9	072– 23 9	RCL • 1	101– 24 .1
h 1/x	073– 25 2	×	102– 61
g x^2	074– 15 3	RCL 8	103– 24 8
RCL 2	075– 24 2	÷	104– 71
×	076– 61	4	105– 4
RCL 6	077– 24 6	•	106– 73
÷	078– 71	6	107– 6
RCL 3	079– 24 3	7	108– 7
+	080– 51	×	109– 61
2	081– 2	1	110– 1
×	082– 61	+	111– 51
RCL 1	083– 24 1	f LN	112– 14 1
RCL 5	084– 24 5	RCL • 2	113– 24 .2
÷	085– 71	×	114– 61
x⇄y	086– 21	+	115– 51

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
CHS	116- 32	+	118- 51
RCL • 0	117- 24 .0	h RTN	119- 25 12

REGISTERS			I v_{i-1}
R_0 v	R_1 ΔP	R_2 L	R_3 $K_T/4$
R_4 ν	R_5 ρ	R_6 D	R_7 ϵ
R_8 R_e	R_9 $1/\sqrt{f}$, f	$R_{.0}$ $1/\sqrt{f_0}$	$R_{.1}$ D/ϵ
$R_{.2}$ 1.737	$R_{.3}$ Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store knowns:			
2a	Store either v or P:	v	STO 0	
		P	STO 1	
2b	Store L:	L	STO 2	
2c	Store $K_T/_4$	K_T	ENTER↑ 4	
			÷ STO 3	
2d	Store ν :	ν	STO 4	
2e	Store ρ:	ρ	STO 5	
2f	Store D:	D	STO 6	
2g	Store ϵ:	ϵ	STO 7	
3	Solve for the unknown:			
	For v		A	v
	or, ΔP		B	ΔP
4	**Optional:** To find R_e		RCL 8	R_e
5	**Optional:** To find f.		RCL 9	f
6	To change any known go to any			
	part of step 2.			

Example 1:

A heat exchanger has 20, 3 meter tube passes (60 m of pipe) with 180 degree bends connecting each pair of tubes (from table 1, $K_T = 10 \times 1.6$). The fluid is water (viscosity $\mu = \nu\rho = 9.3 \times 10^{-4}$ kg/m-s, $\rho = 10^3$ kg/m^3). The surface roughness is 3×10^{-4}m and the diameter is 2.54×10^{-2}m. If the fluid velocity is 3.05m/s what is the pressure loss? What is the Reynolds number? What is the Fanning friction factor?

Keystrokes:	**Display:**	
3.05 [STO] 0		
60 [STO] 2		
16 [ENTER↑] 4 [÷] [STO] 3		
9.3 [EEX] 4 [CHS]		
[ENTER↑] [EEX] 3 [STO] 5		
[÷] [STO] 4		
2.54 [EEX] 2 [CHS]		
[STO] 6		
3 [EEX] 4		
[CHS] [STO] 7 [B]	521,907.5486	(P, N/m^2)
[RCL] 8	83,301.0753	(Re)
[RCL] 9	0.0102	(f)

Example 2:

For the system shown, what is the volume flow rate?



$$K_T = K_0 + K_1 + K_2$$
$$= 1 + .4 + 0.85 = 2.25$$

Keystrokes:	Display:	
78 [ENTER↑]		
32.16 [×]		
62.4 [STO] 5 [×]		
[STO] 1	156,529.1520	(ΔP, lb/ft^2)
250 [STO] 2		
2.25 [ENTER↑] 4 [÷]		
[STO] 3 [EEX] [CHS]		
5 [STO] 4		
.25 [STO] 6		
3.33 [EEX] [CHS]		
6 [STO] 7 [A]	17.7762	(v, ft/sec)

(Note: this calculation will take from 2:15 to 2:30 min.)

Calculate volume flow rate (v x Area).

.25 [ENTER↑] 2 [÷]		
[g] [x^2] [h]		
[π] [×] [×]	0.8726	(ft^3/sec)

Mechanical Engineering

Equations of Motion

This program provides an interchangable solution between displacement, final velocity, acceleration, time and initial velocity for an object that undergoes constant acceleration. Given any three known parameters the two unknowns will be calculated. The motion must be linear.

Equations:

$$x = \frac{t(v + v_0)}{2}$$

$$x = vt - \frac{1}{2} at^2$$

$$x = \frac{v^2 - v_0^2}{2a}$$

$$x = v_0 t + \frac{1}{2} at^2$$

$$v = v_0 + at$$

where: x = displacement

v = final velocity

a = acceleration

t = time

v_0 = initial velocity

Remarks:

- Any consistent set of units may be used.
- Displacement, acceleration, and velocity should be considered signed (vector) quantities. For example: if initial velocity and acceleration are in opposite directions, one should be positive and the other negative.
- All equations assume initial displacement, x_0, equals zero.

KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000–
[h] [LBL] [A]	001–25, 13, 11
[RCL] 1	002– 24 1
[g] [x≠0]	003– 15 61
[GTO] 1	004– 22 1
[RCL] 2	005– 24 2
[g] [x≠0]	006– 15 61
[GTO] 0	007– 22 0
[RCL] [f] [I]	008–24, 14, 23
[RCL] 0	009– 24 0
[×]	010– 61
[RCL] 3	011– 24 3
[RCL] 0	012– 24 0
[g] [x^2]	013– 15 3
[×]	014– 61
2	015– 2
[÷]	016– 71
[+]	017– 51
[STO] 1	018– 23 1
[GTO] 2	019– 22 2
[h] [LBL] 0	020–25, 13, 0
[RCL] 3	021– 24 3
[g] [x≠0]	022– 15 61
[GTO] 0	023– 22 0
[RCL] [f] [I]	024–24, 14, 23
[RCL] 2	025– 24 2
[+]	026– 51
2	027– 2
[÷]	028– 71

KEY ENTRY	DISPLAY
[RCL] 0	029– 24 0
[×]	030– 61
[STO] 1	031– 23 1
[GTO] 9	032– 22 9
[h] [LBL] 0	033–25, 13, 0
[RCL] 0	034– 24 0
[g] [x≠0]	035– 15 61
[GTO] 0	036– 22 0
[RCL] 2	037– 24 2
[g] [x^2]	038– 15 3
[RCL] [f] [I]	039–24, 14, 23
[g] [x^2]	040– 15 3
[−]	041– 41
[RCL] 3	042– 24 3
2	043– 2
[×]	044– 61
[÷]	045– 71
[STO] 1	046– 23 1
[GTO] 5	047– 22 5
[h] [LBL] 0	048–25, 13, 0
[RCL] 2	049– 24 2
[RCL] 0	050– 24 0
[×]	051– 61
[RCL] 3	052– 24 3
[RCL] 0	053– 24 0
[g] [x^2]	054– 15 3
[×]	055– 61
2	056– 2
[÷]	057– 71

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[−]	058− 41	[g] [x²]	087− 15 3
[STO] 1	059− 23 1	[+]	088− 51
[GTO] 8	060− 22 8	[f] [√x]	089− 14 3
[h] [LBL] 1	061−25, 13, 1	[RCL] 1	090− 24 1
[RCL] 2	062− 24 2	[ENTER↑]	091− 31
[g] [x≠0]	063− 15 61	[h] [ABS]	092− 25 34
[GTO] 3	064− 22 3	[÷]	093− 71
[RCL] 3	065− 24 3	[×]	094− 61
[g] [x≠0]	066− 15 61	[STO] 2	095− 23 2
[GTO] 0	067− 22 0	[GTO] 5	096− 22 5
[RCL] 1	068− 24 1	[h] [LBL] 2	097−25, 13, 2
[RCL] 0	069− 24 0	[RCL] 1	098− 24 1
[÷]	070− 71	[RCL] 0	099− 24 0
2	071− 2	[÷]	100− 71
[×]	072− 61	[RCL] 3	101− 24 3
[RCL] [f] [I]	073−24, 14, 23	[RCL] 0	102− 24 0
[−]	074− 41	[×]	103− 61
[STO] 2	075− 23 2	2	104− 2
[GTO] 7	076− 22 7	[÷]	105− 71
[h] [LBL] 0	077−25, 13, 0	[+]	106− 51
[RCL] 0	078− 24 0	[STO] 2	107− 23 2
[g] [x≠0]	079− 15 61	[GTO] 8	108− 22 8
[GTO] 2	080− 22 2	[h] [LBL] 3	109−25, 13, 3
[RCL] 1	081− 24 1	[RCL] 3	110− 24 3
[RCL] 3	082− 24 3	[g] [x=0]	111− 15 71
[×]	083− 61	[GTO] 0	112− 22 0
2	084− 2	[GSB] 6	113− 13 6
[×]	085− 61	[GTO] 8	114− 22 8
[RCL] [f] [I]	086−24, 14, 23	[h] [LBL] 0	115−25, 13, 0

KEY ENTRY	DISPLAY
RCL 0	116– 24 0
g x≠0	117– 15 61
GTO 7	118– 22 7
h LBL 9	119–25, 13, 9
RCL 2	120– 24 2
g x^2	121– 15 3
RCL f I	122–24, 14, 23
g x^2	123– 15 3
–	124– 41
RCL 1	125– 24 1
÷	126– 71
2	127– 2
÷	128– 71
STO 3	129– 23 3
h LBL 5	130–25, 13, 5
RCL 1	131– 24 1
RCL 2	132– 24 2
RCL f I	133–24, 14, 23
+	134– 51
÷	135– 71
2	136– 2
×	137– 61
STO 0	138– 23 0
h RTN	139– 25 12
h LBL 6	140–25, 13, 6
RCL 2	141– 24 2
g x^2	142– 15 3
RCL 1	143– 24 1
RCL 3	144– 24 3

KEY ENTRY	DISPLAY
×	145– 61
2	146– 2
×	147– 61
–	148– 41
f √x	149– 14 3
RCL 1	150– 24 1
ENTER↑	151– 31
h ABS	152– 25 34
÷	153– 71
CHS	154– 32
×	155– 61
RCL 2	156– 24 2
+	157– 51
RCL 3	158– 24 3
÷	159– 71
STO 0	160– 23 0
h RTN	161– 25 12
h LBL 7	162–25, 13, 7
RCL 2	163– 24 2
RCL 1	164– 24 1
RCL 0	165– 24 0
÷	166– 71
–	167– 41
RCL 0	168– 24 0
÷	169– 71
2	170– 2
×	171– 61
STO 3	172– 23 3
h LBL 8	173–25, 13, 8

KEY ENTRY	DISPLAY
RCL 2	174– 24 2
RCL 3	175– 24 3
RCL 0	176– 24 0
×	177– 61

KEY ENTRY	DISPLAY
−	178– 41
STO f I	179– 23, 14, 23
h RTN	180– 25 12

REGISTERS			
			I v_0
R_0 t	R_1 x	R_2 v	R_3 a

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store any 3 of the following:			
	Time, t, if known	t	STO 0	
	Displacement, x, if known	x	STO 1	
	Final velocity, v, if known	v	STO 2	
	Acceleration, a, if known	a	STO 3	
	Initial velocity, v_0, if known	v_0	STO f I	
3	Clear the 2 unknowns:			
	If t is unknown	0	STO 0	
	If x is unknown	0	STO 1	
	If v is unknown	0	STO 2	
	If a is unknown	0	STO 3	
	If v_0 is unknown	0	STO f I	
4	Calculate the unknowns.		A	
5	Recall the desired values:			
	t		RCL 0	t
	x		RCL 1	x
	v		RCL 2	v
	a		RCL 3	a
	v_0		RCL f I	v_0

Example 1:

An automobile accelerates for 4 seconds from a speed of 35 mph and covers a distance of 264 feet. What is the acceleration in ft/sec²? If the acceleration continues to be constant, what distance is covered in the next second?

Keystrokes:	Display:	
264 [STO] 1		
35 [ENTER↑]		
5280 [×]		
3600 [÷] [STO] [f] [I]	51.3333	(v_0, ft/sec)
4 [STO] 0		
0 [STO] 2		
[STO] 3 [A]		
[RCL] 3	7.3333	(a, ft/sec²)
5 [STO] 0		
0 [STO] 1		
[STO] 2 [A]		
[RCL] 1	348.3333	($x_{(t+1)}$, ft)
264 [−]	84.3333	($x_{(t+1)} - x_{(t)}$, ft)

Example 2:

An airplaine's take-off velocity is 125 mph. Assume a constant acceleration of 15 ft/sec². How much runway length, in feet, will be used from start to take-off? How long will it take for the plane to reach take-off velocity?

Keystrokes:	Display:	
125 [ENTER↑]		
5280 [×]		
3600 [÷] [STO] 2	183.3333	(v, ft/sec)
15 [STO] 3		
0 [STO] [f] [I]		($v_0 = 0$)
[STO] 1		
[STO] 0 [A]		
[RCL] 1	1,120.3704	(x, ft)
[RCL] 0	12.2222	(t, sec)

Kinetic Energy

This program calculates an interchangeable solution among the variables weight (or mass), velocity, and kinetic energy, for an object moving at constant velocity. The program operates in either English or metric units. For metric units, any consistent set of units may be used; the quantity mass must be used. For English units, the energy must be in foot-pounds, the velocity in feet per second, and the quantity weight in pounds.

Equations:

English

$$K.E. = \frac{W}{2g} v^2$$

Metric

$$K.E. = \frac{1}{2} m v^2$$

where: $K.E.$ = Kinetic energy

W = Weight (lb)

m = Mass (kg, g)

v = Velocity

g = Acceleration due to gravity = 32.17398 ft/sec^2

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL A	001–25, 13, 11
2	002– 2
STO 0	003– 23 0
h RTN	004– 25 12
h LBL B	005–25, 13, 12
6	006– 6
4	007– 4
•	008– 73
3	009– 3
4	010– 4
7	011– 7
9	012– 9
6	013– 6
STO 0	014– 23 0
h RTN	015– 25 12
h LBL 1	016–25, 13, 1
RCL 2	017– 24 2
RCL 3	018– 24 3
g x^2	019– 15 3
×	020– 61
RCL 0	021– 24 0

KEY ENTRY	DISPLAY
+	023– 71
STO 1	023– 23 1
h RTN	024– 25 12
h LBL 2	025–25, 13, 2
RCL 1	026– 24 1
RCL 3	027– 24 3
g x^2	028– 15 3
+	029– 71
RCL 0	030– 24 0
×	031– 61
STO 2	032– 23 2
h RTN	033– 25 12
h LBL 3	034–25, 13, 3
RCL 1	035– 24 1
RCL 2	036– 24 2
+	037– 71
RCL 0	038– 24 0
×	039– 61
f $\sqrt{x}$	040– 14 3
STO 3	041– 23 3
h RTN	042– 25 12

REGISTERS			I Unused
R_0 Used	R_1 *K.E.*	R_2 *W* or *m*	R_3 *V*
R_4–$R_{.9}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Chose the system of units:			
	SI		[A]	
	or, English		[B]	
3	Input two of the following:			
	Kinetic Energy	*K.E.*	[STO] 1	
	Weight or mass (SI)	*W* or *m*	[STO] 2	
	Velocity	*v*	[STO] 3	
4	Compute the unknown:			
	For *K.E.*		[GSB] 1	*K.E.*
	For *W* or *m*		[GSB] 2	*W* or *m*
	For *v*		[GSB] 3	*v*
5	For a new case go to step 2 or 3.			

Example:

The slider of a slider-crank mechanism is used to punch holes in a slab of metal. It is found that the work required to punch a hole is 775 ft-lb. If the slider weighs 5 lb. 4 oz., how fast must it be moving when it strikes the metal?

Keystrokes:	Display:	
[B]		
775 [STO] 1		
5 [ENTER↑]		
4 [ENTER↑]		
16 [÷] [+] [STO] 2		
[GSB] 3	97.4627	(ft/sec)

Structural Analysis

Mohr Circle for Stress

Given the state of stress on an element, the principal stresses and their orientation can be found. The maximum shear stress and its orientation can also be found.



Stress State

Principal Stresses

Maximum Shear Stress

Equations:

$$s_{smax} = \sqrt{\left(\frac{s_x - s_y}{2}\right)^2 + s_{xy}^2}$$

$$s_1 = \frac{s_x + s_y}{2} + s_{smax}$$

$$s_2 = \frac{s_x + s_y}{2} - s_{smax}$$

$$\theta = ½ \tan^{-1}\left(\frac{2s_{xy}}{s_x - s_y}\right)$$

$$\theta_s = ½ \tan^{-1} -\left(\frac{s_x - s_y}{2s_{xy}}\right)$$

where:

s_{smax} is the maximum shear stress;

s_1 and s_2 are the principal normal stresses;

θ is the angle of rotation from the principal axis to the original axis;

θ_s is the angle of rotation from the axis of maximum shear stress to the original axis;

s_x is the stress in the x direction;

s_y is the stress in the y direction;

s_{xy} is the shear stress on the element.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000–	[x⇄y]	020– 21
[h] [LBL] [A]	001–25, 13, 11	[RCL] 1	021– 24 1
[ENTER↑]	002– 31	[RCL] 2	022– 24 2
[g] [R↓]	003– 15 22	[−]	023– 41
[STO] 3	004– 23 3	[R/S]	024– 74
[g] [R↓]	005– 15 22	[x⇄y]	025– 21
[x⇄y]	006– 21	2	026– 2
[STO] 1	007– 23 1	[÷]	027– 71
[x⇄y]	008– 21	[R/S]	028– 74
[STO] [+] 1	009–23, 51, 1	[RCL] 4	029– 24 4
[−]	010– 41	[RCL] 3	030– 24 3
2	011– 2	[÷]	031– 71
[STO] [÷] 1	012–23, 71, 1	[CHS]	032– 32
[÷]	013– 71	[g] [TAN⁻¹]	033– 15 9
[STO] 4	014– 23 4	2	034– 2
[g] [→P]	015– 15 4	[÷]	035– 71
[STO] 2	016– 23 2	[R/S]	036– 74
[RCL] 1	017– 24 1	[RCL] 2	037– 24 2
[+]	018– 51	[h] [RTN]	038– 25 12
[R/S]	019– 74		

REGISTERS			I Unused
R_0 Unused	R_1 $(s_x + s_y)/2$	R_2 s_{smax}	R_3 s_{xy}
R_4 $(s_x - s_y)/2$	R_5–$R_{.9}$ Unused		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input the following:			
	Stress in the x direction	s_x	ENTER↑	
	(negative for compression)			
	Stress in the y direction	s_y	ENTER↑	
	(negative for compression)			
	Shear stress	s_{xy}		
3	Compute the following:			
	First principal stress		A	s_1
	Second principal stress		R/S	s_2
	Angle of rotation (principal)		R/S	θ
	Angle of rotation (shear)		R/S	θ_s
	Maximum shear stress		R/S	s_{smax}
	Note: Do not disturb the			
	stack during step 3.			
4	For a new case, go to step 2.			

Example:



Keystrokes:	**Display:**	
25000 [ENTER↑]		
5000 [CHS] [ENTER↑]		
4000 [A]	25,524.1747	(s_1, psi)
[R/S]	–5,524.1747	(s_2, psi)
[R/S]	7.4657	(θ, deg)
[R/S]	–37.5343	(θ_s, deg)
[R/S]	15,524.1747	(s_{smax}, psi)

Simply Supported Beams

This program calculates deflection, slope, moment and shear at any specified point along a simply supported beam of uniform cross section. Distributed loads, point loads, applied moments or combinations of all three may be modeled. By using the principle of superposition, complicated beams with multiple point loads, and multiple applied moments can be analyzed.

Equations:



$$y = y_1 + y_2 + y_3 \text{ (total deflection)}$$

$$y_1 = \frac{P(\ell - a)x}{6EI\ell}\left[x^2 + (\ell - a)^2 - \ell^2\right]^* \text{ (deflection due to point load)}$$

$$y_2 = \frac{-Wx}{24EI\ell}\left[\ell^3 + x^2(x-2\ell)\right] \text{ (distributed load)}$$

$$y_3 = \frac{-Mx}{EI\ell}\left[c\ell - \frac{x^2}{6} - \frac{\ell^2}{3} - \frac{c^2}{2}\right]^{**} \text{ (applied moment)}$$

$$\theta = \theta_1 + \theta_2 + \theta_3 \text{ (total slope)}$$

$$\theta_1 = \frac{P(\ell - a)}{6EI\ell}\left[3x^2 + (\ell - a)^2 - \ell^2\right]^{*} \text{ (slope due to point load)}$$

$$\theta_2 = -\frac{W}{24EI}\left[\ell^3 + x^2(4x - 6\ell)\right] \text{ (distributed load)}$$

$$\theta_3 = \frac{-M}{EI}\left[c - \frac{x^2}{2\ell} - \frac{\ell}{3} - \frac{c^2}{2\ell}\right]^{**} \text{ (applied moment)}$$

$$M_x = M_{x1} + M_{x2} + M_{x3} \text{ (total moment)}$$

$$M_{x1} = \frac{P(\ell - a)x}{\ell}^{*} \text{ (moment due to point load)}$$

$$M_{x2} = -\frac{Wx}{2}\left[x - \ell\right] \text{ (distributed load)}$$

$$M_{x3} = \frac{Mx}{\ell}^{**} \text{ (applied moment)}$$

$$V = V_1 + V_2 + V_3 \text{(total shear)}$$

$$V_1 = \frac{P(\ell - a)}{\ell}^{*} \text{(shear due to point load)}$$

$$V_2 = W\left(\frac{\ell}{2} - x\right) \text{ (distributed load)}$$

$$V_3 = \frac{M}{\ell} \text{ (applied moment)}$$

where:

y is the deflection at a distance x from the left support;

θ is the slope (change in y per change in x) at x;

M_x is the moment at x;

V is the shear at x;

I is the moment of inertia of the beam;

E is the modulus of elasticity of the beam;

ℓ is the length of the beam;

P is a concentrated load;

W is the uniformly distributed load with dimensions of force per unit length;

M is an applied moment;

a is the distance from the left support to the point load;

c is the distance to the applied moment.

* If x is great than a, $(\ell - a)$ is replaced by $-a$ and x is replaced by $(x - \ell)$.

** If x is greater than c, x is replaced by $(x - \ell)$ and c is replaced by $(\ell - c)$.

Remarks:

Deflections must not significantly alter the geometry of the problem. Beams must be of constant cross section for deflection and slope equations to be valid. Stresses must be in the elastic region.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL A	001–25, 13, 11
STO 4	002– 23 4
RCL 1	003– 24 1
ENTER↑	004– 31
×	005– 61
h LST x	006– 25 0
×	007– 61
RCL 4	008– 24 4
h F? 0	009–25, 71, 0
4	010– 4
h F? 0	011–25, 71, 0
×	012– 61
RCL 1	013– 24 1
2	014– 2
×	015– 61
h F? 0	016–25, 71, 0
3	017– 3
h F? 0	018–25, 71, 0
×	019– 61
−	020– 41
RCL 4	021– 24 4
g x^2	022– 15 3
×	023– 61
+	024– 51
R/S	025– 74
×	026– 61
2	027– 2
4	028– 4

KEY ENTRY	DISPLAY
÷	029– 71
RCL 4	030– 24 4
x⇄y	031– 21
×	032– 61
h F? 0	033–25, 71, 0
h LST x	034– 25 0
CHS	035– 32
GSB 1	036– 13 1
RCL 5	037– 24 5
g x^2	038– 15 3
h F? 0	039–25, 71, 0
3	040– 3
h F? 0	041–25, 71, 0
×	042– 61
RCL f I	043–24, 14, 23
g x^2	044– 15 3
+	045– 51
RCL 1	046– 24 1
g x^2	047– 15 3
−	048– 41
×	049– 61
6	050– 6
÷	051– 71
GSB 2	052– 13 2
RCL 5	053– 24 5
g x^2	054– 15 3
RCL 1	055– 24 1
÷	056– 71
6	057– 6

KEY ENTRY	DISPLAY
÷	058– 71
h F? 0	059–25, 71, 0
3	060– 3
h F? 0	061–25, 71, 0
×	062– 61
RCL 1	063– 24 1
3	064– 3
÷	065– 71
+	066– 51
RCL f I	067–24, 14, 23
g x^2	068– 15 3
2	069– 2
÷	070– 71
RCL 1	071– 24 1
÷	072– 71
+	073– 51
RCL f I	074–24, 14, 23
−	075– 41
×	076– 61
RCL 1	077– 24 1
×	078– 61
RCL 6	079– 24 6
+	080– 51
RCL 0	081– 24 0
÷	082– 71
h RTN	083– 25 12
h LBL B	084–25, 13, 12
STO 4	085– 23 4
2	086– 2

KEY ENTRY	DISPLAY
÷	087– 71
RCL 1	088– 24 1
h F? 0	089–25, 71, 0
2	090– 2
h F? 0	091–25, 71, 0
÷	092– 71
RCL 4	093– 24 4
−	094– 41
R/S	095– 74
×	096– 61
×	097– 61
h F? 0	098–25, 71, 0
h LST X	099– 25 0
GSB 1	100– 13 1
GSB 2	101– 13 2
RCL 6	102– 24 6
+	103– 51
h RTN	104– 25 12
h LBL 1	105–25, 13, 1
STO 6	106– 23 6
RCL 1	107– 24 1
RCL 2	108– 24 2
−	109– 41
STO f I	110–23, 14, 23
RCL 4	111– 24 4
STO 5	112– 23 5
RCL 2	113– 24 2
f x>y	114– 14 51
GTO 0	115– 22 0

KEY ENTRY	DISPLAY
RCL 2	116– 24 2
CHS	117– 32
STO f I	118– 23, 14, 23
RCL 1	119– 24 1
STO − 5	120– 23, 41, 5
h LBL 0	121– 25, 13, 0
R/S	122– 74
RCL f I	123– 24, 14, 23
×	124– 61
RCL 1	125– 24 1
÷	126– 71
h F? 0	127– 25, 71, 0
h RTN	128– 25 12
RCL 5	129– 24 5
×	130– 61
h RTN	131– 25 12
h LBL 2	132– 25, 13, 2
STO + 6	133– 23, 51, 6
RCL 4	134– 24 4
STO 5	135– 23 5
RCL 3	136– 24 3
STO f I	137– 23, 14, 23
f x>y	138– 14 51
GTO 0	139– 22 0
RCL 4	140– 24 4
RCL 1	141– 24 1
−	142– 41
STO 5	143– 23 5
RCL 1	144– 24 1
RCL 3	145– 24 3
−	146– 41
STO f I	147– 23, 14, 23
h LBL 0	148– 25, 13, 0
R/S	149– 74
RCL 1	150– 24 1
÷	151– 71
h F? 0	152– 25, 71, 0
h RTN	153– 25 12
RCL 5	154– 24 5
×	155– 61
h RTN	156– 25 12

REGISTERS			I $\ell-a$, $-a$; c, $\ell-c$
R_0 EI	R_1 ℓ	R_2 a	R_3 c
R_4 x	R_5 x, $x-\ell$	R_6 Sum	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Set display format.		[f] [ENG] 3	
3	Store beam constants:			
	Moment of inertia	I	[ENTER↑]	
	Modulus of elasticity	E	[×] [STO] 0	
	Length	ℓ	[STO] 1	
4	If applicable, store load			
	location (s):			
	Location of point load	a	[STO] 2	
	Location of applied moment	c	[STO] 3	
5	Calculate any or all of the			
	following at the point of			
	interest, x: (0 must be input			
	for the distributed load, W,			
	point load, P, and applied			
	moment, M, if any of those are			
	not applicable to the beam in			
	question.)			
5a	Deflection:	x	[h] [CF] 0 [A]	
		W	[R/S]	
		P	[R/S]	
		M	[R/S]	y
5b	Slope:	x	[h] [SF] 0 [A]	
		W	[R/S]	
		P	[R/S]	
		M	[R/S]	θ

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
5c	Moment:	x	[h] [CF] 0 [B]	
		W	[R/S]	
		P	[R/S]	
		M	[R/S]	M_x
5d	Shear:	x	[h] [SF] 0 [B]	
		W	[R/S]	
		P	[R/S]	
		M	[R/S]	V
6	For a new beam go to step 3.			
7	For different load locations			
	go to step 4.			
8	For different position, x,			
	go to step 5.			

Example 1:

Find the deflection, slope, internal moment and shear at a distance of 24 inches for the beam below. Neglect the weight of the beam.



Keystrokes:	**Display:**		
[f] [ENG] 3			
.92 [ENTER↑]			
30 [EEX] 6 [×] [STO] 0			
72 [STO] 1			
40 [STO] 3			
24 [h] [CF] 0 [A]			
0 [R/S]			
0 [R/S]			
10000 [R/S]	-30.92	-03	(y_{24})
24 [h] [SF] 0 [A]			
0 [R/S]			
0 [R/S]			
10000 [R/S]	-322.1	-06	(θ_{24})
24 [h] [CF] 0 [B]			
0 [R/S]			
0 [R/S]			
10000 [R/S]	3.333	03	(M_{24})
24 [h] [SF] 0 [B]			
0 [R/S]			
0 [R/S]			
10000 [R/S]	138.9	00	(V_{24})

Example 2:

What is the slope of the beam below at $x = 38$ inches?



$E = 30 \times 10^6$ psi
$I = 1.30$ in^4

Keystrokes:	**Display:**		
1.30 [ENTER↑]			
30 [EEX] 6 [×] [STO] 0			
50 [STO] 1			
44 [STO] 2			
38 [h] [SF] 0 [A]			
25 [R/S]			
1000 [R/S]			
0 [R/S]	3.327	-03	(θ_{38}, in/in)

Example 3:

What is the total moment at the center of the beam below. (It is not necessary to know E or I, since these do not show up in the moment equations).



First solve for the effect of P_2 by itself:

Keystrokes:	**Display:**		
70 [STO] 1			
50 [STO] 2			
35 [h] [CF] 0 [B]			
0 [R/S]			
1000 [R/S]			
0 [R/S]	10.00	03	(in-lb)

Now solve for the effect of the distributed load, P_1, and M and add to it the effect of P_2. This is the final answer assuming superposition is valid.

Keystrokes:	**Display:**		
20 [STO] 2			
70 [STO] 3			
35 [h] [CF] 0 [B]			
37 [R/S]			
400 [R/S]			
10000 [CHS] [R/S]	21.66	03	(in-lb)
10 [EEX] 3 [+]	31.66	03	(M, in-lb)

Section Properties

The properties of arbitrarily shaped sections which are composed of retangles can be evaluated using this program.

The program calculates the area of the section, the centroid of the area, the moments of inertia about any specified set of axes, the polar moment of inertia about the specified axis, the moments of inertia about an axis translated to the centroid, the moments of inertia of the principal axis, and the rotation angle between the translated axis and the principal axis.

Equations:

$$A_{si} = \Delta x_i \, \Delta y_i$$

$$A = A_{s1} + A_{s2} + A_{s3} + \ldots + A_{sn}$$

$$\bar{x} = \frac{\sum_{i=1}^{n} x_{oi} A_{si}}{A} \qquad \bar{y} = \frac{\sum_{i=1}^{n} y_{oi} A_{si}}{A}$$

$$I_x = \sum_{i=1}^{n} \left(y_{oi}^2 + \frac{\Delta y_i^2}{12} \right) A_{si}$$

$$I_y = \sum_{i=1}^{n} \left(x_{oi}^2 + \frac{\Delta x_i^2}{12} \right) A_{si}$$

$$J = I_x + I_y$$

$$I_{xy} = \sum_{i=1}^{n} x_{oi}\, y_{oi}\, A_{si}$$

$$I_{\bar{x}} = I_x - A\bar{y}^2$$

$$I_{\bar{x}\bar{y}} = I_{xy} - A\bar{x}\bar{y}$$

$$I_{\bar{y}} = I_y - A\bar{x}^2$$

$$\phi = \frac{1}{2} \tan^{-1} \frac{-2 I_{\bar{x}\bar{y}}}{I_{\bar{x}} - I_{\bar{y}}}$$

where:

Δx_i is the width of a rectangular element;

Δy_i is the height of a rectangular element;

A_{si} is the area of an element;

A is the total area of the section;

$\bar{x}$ is the x coordinate of the centroid;

$\bar{y}$ is the y coordinate of the centroid;

x_{oi} is the x coordinate of the centroid of an element;

y_{oi} is the y coordinate of the centroid of an element;

I_x is the moment of inertia about the x-axis;

I_y is the moment of inertia about the y-axis;

J is the moment of inertia about the origin;

I_{xy} is the product of inertia;

$I_{\bar{x}}$ is the moment of inertia about the x-axis translated to the centroid;

$I_{\bar{y}}$ is the moment of inertia about the y-axis translated to the centroid;

$I_{\bar{x}\bar{y}}$ is the product of inertia about the translated axis;

ϕ is the angle between the translated axis and the principal axis;

Reference:

Wojciechowski, Felix; ''Properties of Plane Cross Sections''; *Machine Design;* p. 105, Jan 22, 1976.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000–
h LBL A	001–25, 13, 11
STO 4	002– 23 4
g R↓	003– 15 22
STO 3	004– 23 3
g R↓	005– 15 22
STO 2	006– 23 2
x⇄y	007– 21
STO 1	008– 23 1
×	009– 61
STO • 3	010– 23 .3
ENTER↑	011– 31
STO + 0	012–23, 51, 0
RCL 3	013– 24 3
×	014– 61
STO + 5	015–23, 51, 5
g R↓	016– 15 22
RCL 4	017– 24 4
×	018– 61
STO + 6	019–23, 51, 6
RCL 2	020– 24 2
g x^2	021– 15 3
1	022– 1
2	023– 2
÷	024– 71
RCL 4	025– 24 4
g x^2	026– 15 3
+	027– 51
RCL • 3	028– 24 .3

KEY ENTRY	DISPLAY
×	029– 61
STO + 7	030–23, 51, 7
RCL 1	031– 24 1
g x^2	032– 15 3
1	033– 1
2	034– 2
÷	035– 71
RCL 3	036– 24 3
g x^2	037– 15 3
+	038– 51
RCL • 3	039– 24 .3
×	040– 61
STO + 8	041–23, 51, 8
RCL 3	042– 24 3
RCL 4	043– 24 4
×	044– 61
RCL • 3	045– 24 .3
×	046– 61
STO + 9	047–23, 51, 9
h RTN	048– 25 12
h LBL B	049–25, 13, 12
GSB 1	050– 13 1
R/S	051– 74
GSB 0	052– 13 0
R/S	053– 74
RCL 0	054– 24 0
R/S	055– 74
RCL 7	056– 24 7
R/S	057– 74

KEY ENTRY	DISPLAY
RCL 8	058– 24 8
R/S	059– 74
RCL 9	060– 24 9
R/S	061– 74
RCL 7	062– 24 7
GSB 0	063– 13 0
g x^2	064– 15 3
RCL 0	065– 24 0
×	066– 61
−	067– 41
STO • 0	068– 23 .0
R/S	069– 74
RCL 8	070– 24 8
GSB 1	071– 13 1
g x^2	072– 15 3
RCL 0	073– 24 0
×	074– 61
−	075– 41
STO • 1	076– 23 .1
R/S	077– 74
RCL 9	078– 24 9
GSB 1	079– 13 1
GSB 0	080– 13 0
×	081– 61
RCL 0	082– 24 0

KEY ENTRY	DISPLAY
×	083– 61
−	084– 41
STO • 2	085– 23 .2
R/S	086– 74
RCL • 2	087– 24 .2
2	088– 2
×	089– 61
RCL • 1	090– 24 .1
RCL • 0	091– 24 .0
−	092– 41
+	093– 71
g TAN^{-1}	094– 15 9
2	095– 2
÷	096– 71
h RTN	097– 25 12
h LBL 1	098– 25, 13, 1
RCL 5	099– 24 5
RCL 0	100– 24 0
÷	101– 71
h RTN	102– 25 12
h LBL 0	103– 25, 13, 0
RCL 6	104– 24 6
RCL 0	105– 24 0
÷	106– 71
h RTN	107– 25 12

REGISTERS			I Unused
$R_0\ \Sigma A_{si}$	$R_1\ \Delta x_i$	$R_2\ \Delta y_i$	$R_3\ x_{oi}$
$R_4\ y_{oi}$	$R_5\ \Sigma x_{oi}A_{si}$	$R_6\ \Sigma y_{oi}A_{si}$	$R_7\ \Sigma I_x$
$R_8\ \Sigma I_y$	$R_9\ \Sigma xy$	$R_{.0}\ I\bar{x}$	$R_{.1}\ I_{\bar{y}}$
$R_{.2}\ I_{\overline{xy}}$	$R_{.3}\ A_{si}$		

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Initialize.		[f] [REG]	
3	Input Δx, Δy, x_{oi}, y_{oi}.	Δx	[ENTER↑]	
		Δy	[ENTER↑]	
		x_{oi}	[ENTER↑]	
		y_{oi}	[A]	
4	Repeat step 3 for all sections.			
5	Calculate properties:		[B]	$\bar{x}$
			[R/S]	$\bar{y}$
			[R/S]	A
			[R/S]	I_x
			[R/S]	I_y
			[R/S]	I_{xy}
			[R/S]	$I_{\bar{x}}$
			[R/S]	$I_{\bar{y}}$
			[R/S]	$I_{\overline{xy}}$
			[R/S]	ϕ
6	Go to step 2 for a new case.			

Example 1:

Given the rectangle below, find $\bar{x}$, $\bar{y}$, A, I_x, I_y, I_{xy}, $I_{\bar{x}}$, $I_{\bar{y}}$, $I_{\bar{x}\bar{y}}$ and ϕ.



TABLE OF INPUTS

Section	Δx	Δy	x_{oi}	y_{oi}
1	3	5	1.5	2.5

Keystrokes:	Display:	
[f] CLEAR [REG]		
3 [ENTER↑] 5 [ENTER↑]		
1.5 [ENTER↑]		
2.5 [A] [B]	1.5000	($\bar{x}$)
[R/S]	2.5000	($\bar{y}$)
[R/S]	15.0000	(A)
[R/S]	125.0000	(I_x)
[R/S]	45.0000	(I_y)
[R/S]	56.2500	(I_{xy})
[R/S]	31.2500	($I_{\bar{x}}$)
[R/S]	11.2500	($I_{\bar{y}}$)
[R/S]	0.0000	($I_{\bar{x}\bar{y}}$)
[R/S]	0.0000	(ϕ)

Example 2:

Calculate the section properties for the beam shown below.



Section	Δx	Δy	x_{oi}	y_{oi}
1	10	2	6	1
2	1	14	0.5	7
3	15	1	8.5	13.5

Keystrokes:	Display:	
[f] CLEAR [REG]		
10 [ENTER↑] 2 [ENTER↑]		
6 [ENTER↑] 1 [A]		
1 [ENTER↑] 14 [ENTER↑]		
.5 [ENTER↑] 7 [A]		
15 [ENTER↑] 1 [ENTER↑]		
8.5 [ENTER↑]		
13.5 [A] [B]	5.1939	($\bar{x}$)
[R/S]	6.5408	($\bar{y}$)
[R/S]	49.0000	(A)
[R/S]	3,676.3333	(I_x)
[R/S]	2,256.3333	(I_y)
[R/S]	1,890.2500	(I_{xy})
[R/S]	1,580.0017	($I_{\bar{x}}$)
[R/S]	934.4915	($I_{\bar{y}}$)
[R/S]	225.6122	($I_{\bar{x}\bar{y}}$)
[R/S]	–17.4772	(ϕ)

Static Equilibrium at a Point

This program calculates the two reaction forces necessary to balance a given two-dimensional force vector. The direction of the reaction forces may be specified as a vector of arbitrary length or by Cartesian coordinates using the point of force application as the origin.



Equations:

$$R_1 \cos \theta_1 + R_2 \cos \theta_2 = -F \cos \phi$$
$$R_1 \sin \theta_1 + R_2 \sin \theta_2 = -F \sin \phi$$

where:

F is the known force;

ϕ is the direction of the known force;

R_1 is one reaction force;

θ_1 is the direction of R_1;

R_2 is the second reaction force;

θ_2 is the direction of R_2.

The coordinates x_1 and y_1 are referenced from the point where F is applied to the end of the member along which R_1 acts; x_2 and y_2 are the coordinates referenced from the point where F is applied to the end of the member along which R_2 acts.

Remarks:

This program assumes the calculator is set in DEG mode.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000–	[STO] 5	025– 23 5
[h] [LBL] 1	001– 25, 13, 1	[RCL] 4	026– 24 4
[g] [→P]	002– 15 4	[RCL] 3	027– 24 3
[x⇄y]	003– 21	[×]	028– 61
[h] [LBL] 2	004– 25, 13, 2	[RCL] 5	029– 24 5
1	005– 1	[RCL] 2	030– 24 2
[f] [→R]	006– 14 4	[×]	031– 61
[STO] 0	007– 23 0	[−]	032– 41
[x⇄y]	008– 21	[RCL] 1	033– 24 1
[STO] 1	009– 23 1	[RCL] 2	034– 24 2
[h] [RTN]	010– 25 12	[×]	035– 61
[h] [LBL] 3	011– 25, 13, 3	[RCL] 0	036– 24 0
[g] [→P]	012– 15 4	[RCL] 3	037– 24 3
[x⇄y]	013– 21	[×]	038– 61
[h] [LBL] 4	014– 25, 13, 4	[−]	039– 41
1	015– 1	[+]	040– 71
[f] [→R]	016– 14 4	[R/S]	041– 74
[STO] 2	017– 23 2	[h] [LST X]	042– 25 0
[x⇄y]	018– 21	[STO] 6	043– 23 6
[STO] 3	019– 23 3	[RCL] 5	044– 24 5
[h] [RTN]	020– 25 12	[RCL] 0	045– 24 0
[h] [LBL] [A]	021– 25, 13, 11	[×]	046– 61
[f] [→R]	022– 14 4	[RCL] 4	047– 24 4
[STO] 4	023– 23 4	[RCL] 1	048– 24 1
[x⇄y]	024– 21	[×]	049– 61

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[−]	050− 41	[+]	052− 71
[RCL] 6	051− 24 6	[h] [RTN]	053− 25 12

REGISTERS			I Unused
R_0 Cos θ_1	R_1 Sin θ_1	R_2 Cos θ_2	R_3 Sin θ_2
R_4 F cos ϕ	R_5 F sin ϕ	R_6 Used	R_7–$R_{.9}$ Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Define reaction directions as			
	Cartesian coordiantes or as			
	vectors of arbitrary magnitude.			
	(Use the point of force appli-			
	cation as the origin):			
2a	Define direction one in	Y_1	[ENTER↑]	
	rectangular form	X_1	[GSB] 1	
	or in polar form	θ_1	[GSB] 2	
2b	Define direction two in	Y_2	[ENTER↑]	
	rectangular form	X_2	[GSB] 3	
	or in polar form	θ_2	[GSB] 4	
3	Key in known force and			
	compute reactions:			
	Direction	ϕ	[ENTER↑]	
	Magnitude	F	[A]	R_1
			[R/S]	R_2
4	To change force, go to step 3.			
	To change either or both			
	reaction directions, go to			
	step 2.			

Example 1:

Find the reaction forces in the pin-jointed structure shown below.



Keystrokes:	**Display:**	
8 [CHS] [ENTER↑]		
7 [CHS] [GSB] 1		
0 [ENTER↑]		
10 [CHS] [GSB] 3		
90 [CHS] [ENTER↑]		
500 [A]	–664.3841	(R_1)
[R/S]	437.5000	(R_2)

Example 2:

Find the reaction forces for the diagram below:



Keystrokes:	**Display:**	
30 [GSB] 2		
125 [GSB] 4		
240 [ENTER↑]		
100 [A]	90.9770	(R_1)
[R/S]	50.1910	(R_2)

HEWLETT
PACKARD

1000 N.E. Circle Blvd., Corvallis, OR 97330

For additional sales and service information contact your local Hewlett-Packard Sales Office or Call 800/648-4711. Excludes Alaska and Hawaii. (In Nevada call 800/992-5710.)

00034-90035

Printed in U.S.A.